
I. CONSTRUCTION OF POWER CURVES

To construct power curves for each of the parametric and non-parametric retesting strategies,

random standard Normal deviates were generated on an IBM mainframe computer using SAS.  The

background level mean concentration was set to zero, while the alternative mean concentration level

was incremented in steps of ∆=0.5 standardized units above the background level.  At each increment,

5000 iterations of the retesting strategy were simulated; the proportion of iterations indicating

contamination at any one of the wells in the downgradient monitoring network was designated as the

effective power of the retesting strategy (for that ∆ and configuration of background samples and

monitoring wells).

Power values for the EPA Reference Power Curves were not simulated, but represent analytical

calculations based on the non-central t-distribution with non-centrality parameter ∆.  SAS programs

for simulating the effective power of any of the parametric or non-parametric retesting strategies are

presented below.

//********************************************************************;
//* DESCRIPTION:  *** PARAMETRIC SIMULATIONS ***
//*
//* This program produces power curves for 35 different curve
//* simulations (refer to the %LET statements below).  Delta ranges
//* from 0 to 5 by 0.5.  The variable list is as follows for the
//*   input parameters:
//*
//* BG = Background
//* WL = Well
//* TL = Tolerance Limit
//* PL = Prediction Limit
//*
//********************************************************************;
// EXEC SAS
// OUTSAS DD DSN=XXXXXXX.GWT03000.SJA3092.CURVES,
// DISP=OLD
// SYSIN DD *

OPTIONS LS=132 PS=57;
%LET ISTART=1;
%LET CURVENUM=35;
%LET RSEED=2020;
%LET REPEAT=5000;
%LET ITPRINT=1000;

%LET BG1 =24; %LET WL1 =5; %LET TL1 =0.95; %LET PL1 =0.80;
%LET BG2 =24; %LET WL2 =5; %LET TL2 =0.95; %LET PL2 =0.85;
%LET BG3 =8; %LET WL3 =5; %LET TL3 =0.95; %LET PL3 =0.80;
%LET BG4 =8; %LET WL4 =5; %LET TL4 =0.95; %LET PL4 =0.85;
%LET BG5 =24; %LET WL5 =20; %LET TL5 =0.95; %LET PL5 =0.95;
%LET BG6 =24; %LET WL6 =20; %LET TL6 =0.95; %LET PL6 =0.97;
%LET BG7 =8; %LET WL7 =20; %LET TL7 =0.95; %LET PL7 =0.95;
%LET BG8 =8; %LET WL8 =20; %LET TL8 =0.95; %LET PL8 =0.97;
%LET BG9 =24; %LET WL9 =50; %LET TL9 =0.95; %LET PL9 =0.98;
%LET BG10=24; %LET WL10=50; %LET TL10=0.95; %LET PL10=0.99;

%LET BG11=24; %LET WL11=50; %LET TL11=0.99; %LET PL11=0.90;
%LET BG12=24; %LET WL12=50; %LET TL12=0.99; %LET PL12=0.93;
%LET BG13=24; %LET WL13=50; %LET TL13=0.99; %LET PL13=0.94;
%LET BG14=24; %LET WL14=50; %LET TL14=0.98; %LET PL14=0.95;
%LET BG15=24; %LET WL15=50; %LET TL15=0.98; %LET PL15=0.97;
%LET BG16=24; %LET WL16=100; %LET TL16=0.98; %LET PL16=0.97;
%LET BG17=24; %LET WL17=100; %LET TL17=0.98; %LET PL17=0.99;
%LET BG18=24; %LET WL18=100; %LET TL18=0.99; %LET PL18=0.95;
%LET BG19=24; %LET WL19=100; %LET TL19=0.99; %LET PL19=0.97;
%LET BG20=24; %LET WL20=100; %LET TL20=0.99; %LET PL20=0.98;
%LET BG21=8; %LET WL21=20; %LET TL21=0.95; %LET PL21=0.98;
%LET BG22=8; %LET WL22=5; %LET TL22=0.95; %LET PL22=0.90;
%LET BG23=16; %LET WL23=5; %LET TL23=0.95; %LET PL23=0.85;
%LET BG24=16; %LET WL24=5; %LET TL24=0.95; %LET PL24=0.90;
%LET BG25=24; %LET WL25=5; %LET TL25=0.95; %LET PL25=0.90;
%LET BG26=16; %LET WL26=20; %LET TL26=0.95; %LET PL26=0.95;
%LET BG27=16; %LET WL27=20; %LET TL27=0.95; %LET PL27=0.97;
%LET BG28=16; %LET WL28=50; %LET TL28=0.98; %LET PL28=0.95;
%LET BG29=16; %LET WL29=50; %LET TL29=0.98; %LET PL29=0.97;
%LET BG30=16; %LET WL30=50; %LET TL30=0.99; %LET PL30=0.90;
%LET BG31=16; %LET WL31=50; %LET TL31=0.99; %LET PL31=0.92;
%LET BG32=24; %LET WL32=100; %LET TL32=0.98; %LET PL32=0.98;
%LET BG33=16; %LET WL33=100; %LET TL33=0.98; %LET PL33=0.98;
%LET BG34=16; %LET WL34=100; %LET TL34=0.99; %LET PL34=0.95;
%LET BG35=16; %LET WL35=100; %LET TL35=0.99; %LET PL35=0.96;

%MACRO PARSIM;
DATA ITERATE;
*** Set changing simulation variable to common variable names;

BG=&&BG&I;
WL=&&WL&I;
TL=&&TL&I;
PL=&&PL&I;

DO DELTA=0 TO 5 BY 0.5;
*** Initialize TP0, TP1 & TP2 to 0 before entering simulation;

TP0=0;
TP1=0;
TP2=0;

DO J=1 TO &REPEAT;
*** Initialize CNT0, CNT1 & CNT2 to 0;

CNT0=0;
CNT1=0;
CNT2=0;

XB=RANNOR(&RSEED)/SQRT(BG);
SB=SQRT(2*RANGAM(&RSEED,(BG-1)/2)/(BG-1));

PL2=XB+SB*SQRT(1+1/BG)*TINV((1-(1-PL)/2),(BG-1));
PL1=XB+SB*SQRT(1+1/BG)*TINV((1-(1-PL)),(BG-1));
PL0=XB+SB*SQRT(1+1/BG)*TINV((1-(1-TL)),(BG-1));
TLIM=XB+SB*SQRT(1+1/BG)*TINV((1-(1-TL)),(BG-1));

DO K=1 TO WL;
IF K<WL THEN DO;
X1=RANNOR(&RSEED);
X2=RANNOR(&RSEED);
X3=RANNOR(&RSEED);
END;
ELSE DO;
X1=RANNOR(&RSEED)+DELTA;
X2=RANNOR(&RSEED)+DELTA;
X3=RANNOR(&RSEED)+DELTA;

END;
IF X1>TLIM THEN DO;
CNT0=CNT0+1;
IF X2>PL1 THEN CNT1=CNT1+1;
IF X2>PL2 OR X3>PL2 THEN CNT2=CNT2+1;
END;

END;

IF CNT0>0 THEN TP0=TP0+100/&REPEAT;
IF CNT1>0 THEN TP1=TP1+100/&REPEAT;
IF CNT2>0 THEN TP2=TP2+100/&REPEAT;

*** Print iteration information every 100 iterations;
I=&I;
IF MOD(J,&ITPRINT)=0 THEN
   PUT '>>> CURVE ' I ', ITERATION ' J ', ' BG= ', ' WL= ', ' TL= ', '

PL= ', ' DELTA= ', ' TP0= ', ' TP1= ', ' TP2= '<<<';
END;
OUTPUT;
END;
RUN;

DATA OUTSAS.PCURVE&I; SET ITERATE(KEEP=BG WL TL PL TP0 TP1 TP2 DELTA);
RUN;

PROC PRINT DATA=OUTSAS.PCURVE&I;
 FORMAT TP0 TP1 TP2 8.4;
 TITLE1"TEST PRINT OF PARAMETRIC SIMULATION PCURVE&I";
 TITLE2"NUMBER OF ITERATIONS = &REPEAT";
RUN;

%MEND PARSIM;
 %MACRO CURVE;
  %DO I=&ISTART %TO &CURVENUM;
   %PARSIM
  %END;
 %MEND CURVE;
%CURVE

//********************************************************************;
//* DESCRIPTION:  *** NON-PARAMETRIC SIMULATION ***
//*
//* This program produces power curves for 15 different curve
//* simulations (refer to the %LET statements below).  Delta ranges
//* from 0 to 5 by 0.5.  The variable list is as follows for the
//*   input parameters:
//*
//* BG = Background
//* WL = Well
//*
//********************************************************************;
// EXEC SAS
// OUTSAS DD DSN=XXXXXXX.GWT03000.SJA3092.CURVES,DISP=OLD
// SYSIN DD *

OPTIONS LS=132 PS=57;
%LET ISTART=1;
%LET CURVENUM=15;
%LET RSEED=3030;
%LET REPEAT=5000;
%LET ITPRINT=1000;

%LET BG1 =8; %LET WL1 =5;

%LET BG2 =16; %LET WL2 =5;
%LET BG3 =24; %LET WL3 =5;
%LET BG4 =8; %LET WL4 =20;
%LET BG5 =16; %LET WL5 =20;
%LET BG6 =24; %LET WL6 =20;
%LET BG7 =8; %LET WL7 =50;
%LET BG8 =16; %LET WL8 =50;
%LET BG9 =24; %LET WL9 =50;
%LET BG10=8; %LET WL10=100;
%LET BG11=16; %LET WL11=100;
%LET BG12=24; %LET WL12=100;
%LET BG13=32; %LET WL13=100;
%LET BG14=32; %LET WL14=20;
%LET BG15=32; %LET WL15=50;

%MACRO NPARSIM;
DATA ITERATE;
 *** Set changing simulation variable to common variable names;
 BG=&&BG&I;
 WL=&&WL&I;

 DO DELTA=0 TO 5 BY 0.5;
*** Initialize PLx variables to 0 before entering simulation;
PL0=0;
PL1=0;
PL2A=0;
PL2B=0;
PL3A=0;
PL3B=0;

 DO J=1 TO &REPEAT;
*** Initialize CNTx variables to 0;
CNT0=0;
CNT1=0;
CNT2=0;
CNT3=0;
CNT4=0;
CNT5=0;

 DO K=1 TO BG;
TEST=RANNOR(&RSEED);
IF K=1 THEN MAX=TEST;
 ELSE IF TEST>MAX THEN MAX=TEST;

 END;

 DO L=1 TO WL;
IF L<WL THEN DO;
X1=RANNOR(&RSEED);
X2=RANNOR(&RSEED);
X3=RANNOR(&RSEED);
X4=RANNOR(&RSEED);
END;

 ELSE DO;
X1=RANNOR(&RSEED)+DELTA;
X2=RANNOR(&RSEED)+DELTA;
X3=RANNOR(&RSEED)+DELTA;
X4=RANNOR(&RSEED)+DELTA;

END;
IF X1>MAX THEN DO;

CNT0=CNT0+1;
IF X2>MAX THEN CNT1=CNT1+1;
IF X2>MAX & X3>MAX THEN CNT2=CNT2+1;
IF X2>MAX OR X3>MAX THEN CNT3=CNT3+1;
IF X2>MAX & X3>MAX & X4>MAX THEN CNT4=CNT4+1;

IF X2>MAX OR X3>MAX OR X4>MAX THEN CNT5=CNT5+1;
END;

IF CNT0>0 THEN PL0=PL0+100/&REPEAT;
IF CNT1>0 THEN PL1=PL1+100/&REPEAT;
IF CNT2>0 THEN PL2A=PL2A+100/&REPEAT;
IF CNT3>0 THEN PL2B=PL2B+100/&REPEAT;
IF CNT4>0 THEN PL3A=PL3A+100/&REPEAT;
IF CNT5>0 THEN PL3B=PL3B+100/&REPEAT;

*** Print iteration information every X iterations;
I=&I;
IF MOD(J,&ITPRINT)=0 THEN
 PUT '>>> CURVE ' I ', ITERATION ' J ', ' BG= ', ' WL= ', ' DELTA=

', ' PL0= ', ' PL1= ', ' PL2A= ', ' PL2B= ', ' PL3A= ', ' PL3B= '<<<';
END;
OUTPUT;
END;
RUN;

DATA OUTSAS.NCURVE&I; SET ITERATE(KEEP=BG WL PL0 PL1 PL2A PL2B PL3A PL3B DELTA);
RUN;

PROC PRINT DATA=OUTSAS.NCURVE&I;
 FORMAT PL0 PL1 PL2A PL2B PL3A PL3B 8.4;
 TITLE1"TEST PRINT OF NON-PARAMETRIC SIMULATION NCURVE&I";
 TITLE2"NUMBER OF ITERATIONS = &REPEAT";
RUN;

%MEND NPARSIM;
 %MACRO CURVE;
  %DO I=&ISTART %TO &CURVENUM;
   %NPARSIM
  %END;
 %MEND CURVE;
%CURVE

II. PARAMETRIC RETESTING STRATEGIES

